


Assembler BootCamp Plus: Instructions Everyone Can Use

(Created by)

John Dravnieks, IBM Australia
(dravo@au1.ibm.com)

(Presented by Dan Greiner and John Ehrman)



SHARE 118, Atlanta, GA
Friday March 16, 2012
Session 10345

1

Agenda

- Bit shifting
- Single byte operands
- Halfword operands
- Multiple byte operands
- Variable length operands
- Character translation

2


Definitions

- Characters used in instruction mnemonics
 - ▶ **G** - **G**rande - 64-bit operand
 - ▶ **F** - **F**ullword - 32-bit operand
 - ▶ **H** - **H**alfword - 16-bit operand
 - ▶ Single byte operands
 - **B** - **B**yte (signed 8 bit value)
 - **C** - **C**haracter
 - ▶ **L** - **L**ogical - unsigned, or Load and clear
 - ▶ **Y** - 20-bit displacement

3

Definitions

- Parts of a 64-bit register




- **H**=High, **L**= Low, **F**=Fullword

4

Bit shifting

- RS-type format instructions



- R₁ - Source **and** target
- 2nd operand address
 - ▶ **NO** storage reference
 - ▶ Last 6 bits used as shift amount

5

Bit shifting (*continued*)

- Two directions, two types, and two sizes
 - ▶ Left or Right
 - ▶ Logical or Arithmetic
 - ▶ Single or Double register
- 8 mnemonics - **Shift ...**
 - SLA SLDA
 - SLL SLDL
 - SRA SRDA
 - SRL SRDL

6

Bit shifting (*continued*)

- 64-bit register instructions
 - ▶ **NO** 64-bit-register-pair (128-bit) shifts
 - ▶ Single-length: **SLAG**, **SRAG**, **SLLG**, **SRLG**
- Separate source (R_3) and target (R_1) registers
- Example:
SLAG $R_1, R_3, D_2(B_2)$
 - ▶ Shifted contents of R_3 goes into R_1

7

Bit shifting (*continued*)

- Arithmetic shifts:
 - ▶ Sign bit not modified
 - ▶ Right shifts copy sign bit
 - ▶ Left shifts may overflow
 - ▶ Condition code set
- Logical shifts:
 - ▶ No sign bit
 - ▶ Always inserts 0's
 - ▶ Condition code not changed

8

Bit shifting: example 1

- **SRA 5,16**
 - ▶ Object code x'8A50 0010'
 - ▶ c(r5) before x'8001 0000' (sign is propagated)
 - ▶ c(r5) after x'FFFF 8001'
 - ▶ Condition code 1 set (result < 0)
- **SRA 5,7**
 - ▶ Object code x'8A50 0007'
 - ▶ c(r5) after x'FF00 0200'
- **SRA 5,20**
 - ▶ c(r5) after x'FFFF F800'

9

Bit shifting: example 2

- **SRL 5,16**
 - ▶ Object code x'8850 0010'
 - ▶ c(r5) before x'8001 FFFF'
 - ▶ c(r5) after x'0000 8001' (sign not propagated)
- **SRL 5,7**
 - ▶ Object code x'8850 0007'
 - ▶ c(r5) after x'0100 03FF' (3 = 0011)
- **SRL 5,20**
 - ▶ c(r5) after x'0000 0800'

10

Bit shifting: example 3

- **SLA 5,16**
 - ▶ Object code **x'8B50 0010'**
 - ▶ c(r5) before **x'0000 8001'**
 - ▶ c(r5) after **x'0001 0000'**
 - ▶ Condition code 3 set (Overflow)
- **SLA 5,7**
 - ▶ Object code **x'8B50 0007'**
 - ▶ c(r5) after **x'0040 0080'** (CC2, no overflow)
- **SLA 5,30**
 - ▶ c(r5) after **x'4000 0000'** (overflow)

11

Bit shifting: example 4

- **SLL 5,16**
 - ▶ Object code **x'8950 0010'**
 - ▶ c(r5) before **x'0000 8001'**
 - ▶ c(r5) after **x'8001 0000'**
- **SLL 5,7**
 - ▶ Object code **x'8950 0007'**
 - ▶ c(r5) after **x'0040 0080'**
- ▶ **SLL 5,30**
 - ▶ c(r5) after **x'4000 0000'**

12

Bit shifting (*continued*)

■ Rotate Left Single Logical

- ▶ **RLL (G) $R_1, R_3, D_2 (B_2)$**
 - Separate target (R_1) and source (R_3) registers
- ▶ Example: **RLL 7, 8, 12 (0)**
- ▶ Before: $c(R7)=x'????????', c(R8)=x'FEDC0000'$
- ▶ After: $c(R7)=x'C0000FED', c(R8)=x'FEDC0000'$

13

Bit shifting: uses

■ Arithmetic Operations

- ▶ Fast multiplication or division by a power of 2
- ▶ Hashing algorithms

■ Masking

- ▶ In conjunction with Boolean operations
 - Exclusive OR (XOR), OR, AND
- ▶ Extracting data
 - Merged or compressed data

■ Encryption

14

Single byte operands

■ Insert Character

- ▶ $IC \quad R_1, D_2(X_2, B_2)$
- ▶ Copies a single byte from storage into low order byte of R_1
- ▶ **Note:** rest of R_1 register unchanged

■ STore Character

- ▶ $STC \quad R_1, D_2(X_2, B_2)$
- ▶ Copies the low order byte of R_1 into storage

15

Single byte operands: example 1

- $IC \quad 7, 0(0, 11)$
 - ▶ Object text $x'4370 \ B000'$
- R11 points to storage byte containing $x'A5'$
- c(R7) before $x'1234 \ 5678'$
- c(R7) after $x'1234 \ 56A5'$
 - ▶ Remainder of register R7 is unchanged
- Condition code is unchanged

16

Single byte operands

■ Load Logical Character

- ▶ **LL(G)CR** R_1, R_2
- ▶ **LL(G)C** $R_1, D_2(X_2, B_2)$
- ▶ Clears the register and copies a byte from register or storage into low order byte of R_1

■ Load Byte

- ▶ **L(G)BR** R_1, R_2
- ▶ **L(G)B** $R_1, D_2(X_2, B_2)$
- ▶ Single byte from register or storage is sign extended and updates the *entire* register

17

Single byte operands: example 2

- **LLC 7, 0(0, 11)** Load Logical Character
 - ▶ Object text **x'E370 B000 0094'**
- R11 points to storage byte containing **x'A5'**
- c(R7) before **x'1234 5678'**
- c(R7) after **x'0000 00A5'**
 - ▶ Remainder of register R7 is zeroed
- Condition code is unchanged

18

Single byte operands: example 3

- **LB 7,0(0,11)** Load Byte
 - ▶ Object text **x'E370 B000 0076'**
- R11 points to storage byte containing **x'A5'**
- c(R7) before **x'1234 5678'**
- c(R7) after **x'FFFF FFA5'**
 - ▶ Leftmost bit of **x'A5'** extended to left
- Condition code is unchanged

19

Single byte operands: uses

- Translation example (we'll use it again):

```

    ...
    UNPK  STRING(L'String),HEXDATA(L'HEXDATA+1)
* Get data into zoned format.
    LA    3,STRING          Point to STRING.
    LHI   4,1               Load JXLE increment.
    LA    5,L'String-1(,3)  Point at last byte.
LOOP    IC    2,0(,3)       Get next character.
        NILL  2,X'000F'     Remove zone.
        IC    2,table(2)    Use c(R2) as index.
        STC   2,0(,3)       Store "translated" digit.
        JXLE  3,4,LOOP      Loop until finished.
    ...
TABLE   DC    C'0123456789ABCDEF'
```

- The low-order hex digit of each byte referenced by R3 is replaced by its character representation

20

Halfword (two byte) operands

- RX instructions
 - Mnemonic $R_1, D_2(X_2, B_2)$
- Operand 1 is entire R_1 register
 - **STH** ignores high order 16 bits of R_1 , stores only rightmost 16 bits
- Operand 2
 - ▶ Halfword in storage
 - ▶ Signed value - **LH** expands to fullword with sign extension

21

Halfword (two byte) operands (continued)

- | | |
|--------------------------|------------|
| ■ Add Halfword | AH |
| ■ Compare Halfword | CH |
| ■ Load Halfword | LH |
| ■ Multiply Halfword | MH |
| ■ ST ore Halfword | STH |
| ■ Subtract Halfword | SH |

22

Halfword (two byte) operands

(continued)

- Halfword immediate format
 - **Mnemonic** R_1, I_2
where I_2 is a signed 16-bit field in the instruction
- **Add Halfword Immediate** **AHI**
- **Compare Halfword Immediate** **CHI**
- **Load Halfword Immediate** **LHI**
- **Multiply Halfword Immediate** **MHI**

23

Halfword (two byte) operands

(continued)

- Halfword-immediate operands for 64-bit registers:
 - ▶ **AGHI, CGHI, LGHI, MGHI**
 - ▶ **LGH(R)**
- Long displacement facility (instructions with signed 20-bit displacement)
 - ▶ **AHY, CHY, LHY, STHY, SHY**

24

Halfword (two byte) operands

(continued)

- Register-to-register form: **L(G)HR**
 - ▶ Source is in bits 48-63 of 2nd-operand register
- Load Logical form: **LL(G)HR, LL(G)H**
 - ▶ Remainder of 1st-operand register zeroed
- Load Logical Immediate form: **LLlxx**
 - ▶ Source is in bits 16-31 or 16-47 of the instruction
- Insert Immediate form: **llxx**
 - ▶ Remainder of 1st-operand register unchanged
- Where **xx** - **HH, LH, HL, LL** (See slide 4)

25

Halfword operands: example 1

- **LH 0,0(0,12)**
 - ▶ Object text **x'4800 c000'**
- R12 points to storage containing **x'B1A4'**
- c(R0) before **x'FEDC BA98'**
- c(R0) after **x'FFFF B1A4'**
 - ▶ High-order bit of **x'B1A4'** extended to left
- Condition code is unchanged

26

Halfword operands: example 2

- `CH 10,0(0,11)`
 - ▶ Object text `x'49A0 B000'`
- R11 points to storage containing `x'B1A4'`
 - ▶ Expanded internally to `x'FFFF B1A4'`
- If `c(R10) = x'FFFF B1A4'`
 - ▶ Condition code set to 0 (equal)
 - ▶ R10 unchanged
- If `c(R10) = x'0000 B1A4'`
 - ▶ Condition code set to 2 (greater)

27

Halfword operands: example 3

- `CH 10,0(0,11)`
 - ▶ Object text `x'49A0 B000'`
- R11 points to storage containing `x'B1A4'`
- If `c(R10) = x'FFFF A5A5'`
- Resulting Condition Code ?
- Is R10 unchanged?

28

Halfword operands: example 4

- **LLILH** 0, X' **A5A5** '
 - ▶ Load Logical Immediate Low High
 - ▶ Object text X' A50E **A5A5** '
- c(R0) before X' FEDC BA98 '
- c(R0) after X' **A5A5** 0000 '
 - ▶ Remainder of target register is zeroed
- Condition code is unchanged

29

Halfword operands: example 5

- **IILH** 0, X' **A5D6** '
 - ▶ Insert Immediate Low High
 - ▶ Object text X' A502 **A5D6** '
- c(R0) before X' FEDC BA98 '
- c(R0) after X' **A5D6** BA98 '
 - ▶ Remainder of target register is unchanged
- Condition code is unchanged

30

Halfword operands: uses

- Record lengths (DCBLRECL)
 - ▶ V format records: RDWs, BDWs
- Database records
- Small integers

31

Multiple byte operands

- **Insert Characters under Mask**
 - ▶ **ICM** $R_1, \text{Mask}, D_2(B_2)$
 - ▶ Copies 0 to 4 bytes from storage into mask-selected bytes of R_1
 - ▶ Condition code set
 - ▶ **Note:** Unselected bytes unchanged

32

Multiple byte operands (*continued*)

- Mask operand is a 4 bit field
 - ▶ Bits correspond one to one with bytes of register
 - ▶ `B'1001'` refers to the first and last byte
- Storage bytes are contiguous
 - ▶ `ICM 2,B'1010',=X'12345678'`
 - ▶ `c(R2) = X'12??34??'`

33

Multiple byte operands (*continued*)

- **C**ompare **L**ogical Characters under **M**ask
 - ▶ `CLM R1,Mask,D2(B2)`
 - ▶ Compares 0 to 4 contiguous bytes from storage with mask-selected bytes of R₁
 - ▶ Condition code is set
- **S**Tore **C**haracters under **M**ask
 - ▶ `STCM R1,Mask,D2(B2)`
 - ▶ Stores 0 to 4 bytes from selected bytes of R₁ register into contiguous storage bytes

34

Multiple byte operands (*continued*)

- z/Architecture instructions:
 - ▶ CLMY, CLMH
 - ▶ ICMY, ICMH
 - ▶ STCMY, STCMH
- **H** = **H**igh-order 32 bits of 64-bit register
- Long-displacement format (RSY)

35

Multiple byte operands: uses

- `STCM R1, B'0111', D2(R2)`
 - ▶ Stores low-order 24 bits of R₁ into contiguous storage bytes
 - ▶ Historically important use:
 - `STCM R5, B'0111', Label+1`
 - `Label DC X'bits', AL3(address)`
 - DCB address fields
 - CCW address field

36

Multiple byte operands: uses (continued)

- ICM with mask **B'0001'**
 - ▶ Same as IC, but condition code is set
- ICM with mask **B'1111'**
 - ▶ Same as Load, but condition code is set
 - ▶ **ICM 5,B'1111',24(8)** is equivalent to:
 - ▶ **L 5,24(,8)** this
 - ▶ **LTR 5,5** plus this
 - ▶ **NO** index register with ICM

37

Fullword operands

- z/Architecture with extended immediate facility
 - ▶ Load and Test - **LT** (like **L** + **LTR**)
 - ▶ 32-bit **F**ullword **I**mmEDIATE operands:
 - Arithmetic: **AFI, ALFI, SLFI**
 - Logical AND, XOR, OR: **NIHF, NILF, XIHF, XILF, OIHF, OILF**
 - Compare: **CFI, CLFI**
 - Load immediate: **LGFI, LLIHF, LLILF**
 - Insert immediate: **IIHF, IILF**

38

Variable number of operand bytes

- Q: How would we store HLASM symbols, from 1 to 63 bytes long?
- A1: Update MVC instruction in storage?
 - ▶ Reentrancy violation
 - ▶ Difficult to debug
 - ▶ Data / Instruction cache conflicts?
- A2: Use **IC** and **STC** in a loop?
 - ▶ Slow
- A3: Use **EX**ecute instruction!

39

EXecute instruction

- **EX** $R_1, D_2(X_2, B_2)$
- Operand 2 - Address of target instruction
- If R_1 is not general register 0, then low order byte is ORed **internally** with the **second** byte of the target instruction
- The target instruction is then performed
 - ▶ The target instruction in memory is unchanged!

40

EXecute instruction *(continued)*

- Three important points
 - ▶ Operands 1 and 2 are not modified
 - ▶ The operation is a logical OR
 - ▶ When EXecuting variable-length instructions, lengths in object text are one less than actual length
- An example follows

41

EXecute instruction example

- **EX** **R4**,MOVEIT
- **MOVEIT** **MVC TARGET(0),SOURCE**
 - ▶ Object text **x'D200 bddd bddd'**
- **c(R4) = x'1234 5602'**
- Effective object text **x'D202 bddd bddd'**
- So three (3) bytes are moved

42

EXecute instruction: lengths

- R4 in that example holds *machine* length
- If R4 holds *actual* length, then how do we make R4 the machine length (one less)?
 - ▶ Any one of these:
 - ▶ `S R4,=F'1' (or SH R4,=H'1') (?)`
 - ▶ `BCTR R4,0`
 - ▶ `LA R4,255(,R4)`
 - ▶ `AHI R4,-1 (Recommended!)`

43

EXecute instruction: uses

- Often, the target instruction is SS format, like `MVC, CLC, TR` or `TRT`
 - ▶ Only target instructions not allowed are EX & EXRL
- **NOP** (i.e., `BC 0`) can be **EX**ecuted
 - ▶ Use mask of `x'F0'` for unconditional branch
 - ▶ Use other mask for program-specified condition
 - ▶ Target of `BC 15,...` will always branch, regardless of **EX** `R1` field
 - However, bits 12-15 of the target can be modified (e.g., `BCR R2` field)
- Example:
 - ▶ **EX 0,Target_SVC**
 - Allows shared code (Test and Production) to use different SVCs

44

Variable number of operand bytes - Take 2

- Q: How would we store character strings from 1 to 567 bytes long?
- A1: Update instruction in storage (**Bad!**)
 - ▶ Won't work anyway: max length is 256
- A2: Use **IC** and **STC** in a loop?
 - ▶ Even slower
- A3: Use **EX**ecute instruction? (Not bad...)
 - ▶ Loop moving 256 byte chunks and then an **EX**ecuted move at the end (used in old days)
- A4: Use Move Long!

45

Move Long instruction

- **MVCL** R_1, R_2
 - ▶ **MVCL** 4,6 - object text **x'0E46'**
- Operands designate even-odd register pairs:
 - ▶ Even register: operand address
 - ▶ Odd register (even+1): operand length
 - Source length register has **pad character** in high order byte
 - Maximum length is 16MB (24 remaining bits of the odd registers)

46

Move Long instruction (*continued*)

- All 4 registers may be modified
- Sets condition code
- R0 (implying the pair R0 and R1) is valid
 - ▶ Yes, R0 *can* contain an address!
- Clear a block of storage:
 - ▶ `LM 0,3,=A(Block,L'Block,0,0)`
 - ▶ `MVCL 0,2 X'00'` Pad char in R3

47

Compare Logical Long instruction

- `CLCL R1,R2`
 - ▶ `CLCL 4,6` - object text `X'0F46'`
- Same register setup as MVCL
- All 4 registers may be modified - data in storage is NOT modified
- Shorter operand padded with pad character
- Condition code is set

48

CLCL example

■ Example of CLCL usage

- ▶ **LM** **2,3,=A(String1,L'String1)**
 Target addr, length
- ▶ **LM** **0,1,=A(String2,L'String2)**
 Source addr, length
- ▶ **ICM** **1,B'1000',=C' ' Pad byte**
- ▶ **CLCL** **2,0**
- ▶ **BE** **Equal_strings**

49

Extended Move and Compare Long

- Move Long Extended (**MVCLE**)
- Move Long Unicode (**MVCLU**)
- Compare Logical Long Extended (**CLCLE**)
- Compare Logical Long Unicode (**CLCLU**)
 - Lengths can be greater than 16MB
 - Pad character formed from 2nd operand
 - Unicode: 2 bytes per step
 - CC set to 3 if operation is incomplete

50

Extended Move and Compare Long - examples

Compare CLCLE 2,0,X'40' blank pad
 BO Compare CC3 test
 BE Equal_strings

CompUni CLCLU 2,0,X'020'
 BO CompUni CC3 test
 BE Equal_strings

51

Move with Optional Specifications

- **MVCOS** $D_1(B_1), D_2(B_2), R_3$
 - ▶ Set GPR0 to zero
 - ▶ Set R3 operand to **TRUE** length
 - ▶ Moves 0 - 4096 bytes
 - If true length greater than 4096, then 4096 bytes moved and condition code 3 is set
 - Otherwise, true length bytes moved and condition code 0 is set

52

Translation

- Q: How to ensure that character data is in upper case?
- A1: Use the IC/STC code earlier (slide 20) with a new table
- A2: Use TRanslate instruction !

53

TRanslate instruction

- `TR D1(L1,B1),D2(B2) SS format`
- Operand 1 is source and target
- Operand 2 is address of translate table
 - ▶ Usually 256 bytes - depends on data
- `TR STR,Table`
- `STR DC C'Hello, World!'`
- `Table DC C'.....' (See next page)`

54

TRanslate instruction (*continued*)

TABLE addresses a 256 byte table where each data byte is the desired output byte for that offset. For example, this table would translate lower case EBCDIC to upper case EBCDIC.

```
CAPTABLE DS 0CL256 0 1 2 3 4 5 6 7 8 9 A B C D E F
DC XL16'000102030405060708090A0B0C0D0E0F' 00-0F
DC XL16'101112131415161718191A1B1C1D1E0F' 10-1F
DC XL16'202122232425262728292A2B2C2D2E2F' 20-2F
DC XL16'303132333435363738393A3B3C3D3E3F' 30-3F
DC XL16'404142434445464748494A4B4C4D4E4F' 40-4F
DC XL16'505152535455565758595A5B5C5D5E5F' 50-5F
DC XL16'606162636465666768696A6B6C6D6E6F' 60-6F
DC XL16'707172737475767778797A7B7C7D7E7F' 70-7F
DC XL16'80C1C2C3C4C5C6C7C8C98A8B8C8D8E8F' 80-8F
DC XL16'90D1D2D3D4D5D6D7D8D99A9B9C9D9E9F' 90-9F
DC XL16'A0A1E2E3E4E5E6E7E8E9AAABACADAEAF' A0-AF
DC XL16'B0B1B2B3B4B5B6B7B8B9BABBBBCDBEBF' B0-BF
DC XL16'C0C1C2C3C4C5C6C7C8C9CACBCCDCECF' C0-CF
DC XL16'D0D1D2D3D4D5D6D7D8D9DADBDCDDDEDF' D0-DF
DC XL16'E0E1E2E3E4E5E6E7E8E9EAEBECEDEEEF' E0-EF
DC XL16'F0F1F2F3F4F5F6F7F8F9FAFBFCFDFEFF' F0-FF
```

55

TRanslate instruction (*continued*)

- Each byte in operand 1 is used to index into operand 2; that function byte from table replaces the source byte
- **TR STR, TABLE**
 - Single instruction replaces previous five instruction loop (see slide 20)

56

TRanslate instruction - example

■ Translate hex data to printable characters

```
►          UNPK  STRING(L'String+1),HEXDATA(L'HEXDATA+1)
*  Get data into zoned format
          LA     R5,L'String-1          Load machine length
          EX     R5,TR_INST             Perform translation
          ...
TR_INST    TR     STRING(0),TABLE       Executed TRANSLATE
          ...
          ORG    *-240                  Position label
TABLE      DS     0X                    Start of table
          ORG    *+240                  Skip to actual data
          DC     C'0123456789ABCDEF'
```

57

Related instructions

■ Translate and Test

- **TRT** $D_1(L_1, B_1), D_2(B_2)$ Left to right
- **TRTR** $D_1(L_1, B_1), D_2(B_2)$ Right to left

■ Operands not modified

■ Table - operand 1 byte used as index

- If table byte is zero, scan continues
- If non zero, scan stops
 - GR1: Address of operand 1 byte
 - GR2: Test-table byte

58

Related instructions

■ Translate Extended

- ▶ **TRE** R_1, R_2
 - First operand address in register R_1
 - First operand length in register R_1+1
 - Translate table address in register R_2
- ▶ Test byte in GR0
 - Translation stops if it matches source byte
 - Registers updated

59

TRT instruction - example

■ Scan for ASCII (X'20') or EBCDIC (X'40') blanks

- ▶ **SR** $R2, R2$ Clear R2
- ▶ **LA** $R1, \text{STRING} + \text{L'STRING} - 1$ Set R1 to last byte
- ▶ **LA** $R5, \text{L'STRING} - 1$ Load machine length
- ▶ **EX** $R5, \text{TRT_INST}$ Perform scan
- ▶ **JZ** No_Blanks Nothing found (CC 0)
- ▶ **CHI** $R2, \text{X}'20'$ ASCII blank?
- ▶ ...
- ▶ **TRT_INST** $\text{TRT STRING}(0), \text{TABLE}$ Executed TRT
- ▶ **TABLE** **DC** $256\text{X}'00'$ Define 256 byte table
- ▶ **ORG** $\text{TABLE} + \text{X}'20'$ Move to offset X'20'
- ▶ **DC** $\text{X}'20'$ Set non zero
- ▶ **ORG** $\text{TABLE} + \text{X}'40'$ Move to offset X'40'
- ▶ **DC** $\text{X}'40'$ Set non zero
- ▶ **ORG**, Skip to end of TABLE

60

Summary

Many useful instructions!

- Bit shifting
- Single byte operands
- Halfword operands
- Multiple byte operands
- Variable length operands
- Character translation